/*
Copyright (c) 2009, Young Digital Planet SA
All rights reserved.

* Redistribution and use in source and binary forms, with or without 
  modification, are permitted provided that the following conditions are met:
* Redistributions of source code must retain the above copyright notice, this 
  list of conditions and the following disclaimer.
* Redistributions in binary form must reproduce the above copyright notice, 
  this list of conditions and the following disclaimer in the documentation 
  and/or other materials provided with the distribution.
* Neither the name of the Young Digital Planet SA nor the names of its 
  contributors may be used to endorse or promote products derived from this 
  software without specific prior written permission.

THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS IS" 
AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED TO, THE 
IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A PARTICULAR PURPOSE 
ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT HOLDER OR CONTRIBUTORS BE 
LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL, SPECIAL, EXEMPLARY, OR 
CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED TO, PROCUREMENT OF 
SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR PROFITS; OR BUSINESS 
INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF LIABILITY, WHETHER IN 
CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING NEGLIGENCE OR OTHERWISE) 
ARISING IN ANY WAY OUT OF THE USE OF THIS SOFTWARE, EVEN IF ADVISED OF THE 
POSSIBILITY OF SUCH DAMAGE.
*/

package com.yauthor.module.editor.services.colorpicker
{
	import flash.display.DisplayObject;
	import flash.events.IEventDispatcher;
	
	//events
	/**
 	* Dispatched when color picker is loaded and ready to work.
 	*/ 
	[Event(name="cssPalettePickerReady", type="com.yauthor.module.editor.events.CssPalettePickerEvent")]
	
	/**
	 * Dispatched when picker was opened.
	 */
	[Event(name="cssPalettePickerOpen", type="com.yauthor.module.editor.events.CssPalettePickerEvent")]
	
	/**
	 * Dispatched when picker was closed.
	 */
	[Event(name="cssPalettePickerClose", type="com.yauthor.module.editor.events.CssPalettePickerEvent")]
	
	/**
	 * Dispatched when picker property was changed. ( such as selectedColor in example )
	 */
	[Event(name="propertyChange", type="mx.events.PropertyChangeEvent")]
	
	
	/**
	 * Interface for the color picker service.
	 * 'ICssPalettePicker' is a visual component that allows to select color 
	 * from a palette defined in the css the file.
	 * Events dispatched by service are defined in CssPalettePickerEvent class
	 * @see com.yauthor.module.editor.events.CssPalettePickerEvent
	 * 
	 * @example The following code creates instance of ICssPalettePicker and reads properties of a selected color

	<listing version="3.0">
	var colorPicker:ICssPalettePicker = IEditorEnvironment( env ).getService( ICssPalettePicker );
	addChild (colorPicker.view);
	 
		colorPicker.addEventListener( PropertyChangeEvent.PROPERTY_CHANGE, function (e:PropertyChangeEvent):void{
				if ( e.property == "selectedColor" ){
					trace ("selected color code:" + colorPicker.selectedColor);
					trace ("related style property:" + colorPicker.selectedColorStyleProp );
				} 
		} );

	</listing>
	 */  
	 
	public interface ICssPalettePicker extends IEventDispatcher
	{
		
		/**
		 * Method forces color picker to close.    
		 **/ 
		function close():void;
		
		/**
		 * Method forces color picker to open.    
		 **/
		function open():void;
		
		/**
		 * Definition of the selected color.
		 * Note that flex PropertyChangeEvent is being dispatched on 'selectedColor' change.
		 * @return color description object  
		 * @see com.yauthor.module.editor.services.colorpicker.IPalleteColor
		 **/
		function get selectedColor():IPalleteColor;
		
		/**
		 * Method allows to set color selection
		 * Color can be also set by setting 'selectedColorStyleProp' variable.
		 * Note that flex PropertyChangeEvent is being dispatched on 'selectedColor' change.    
		 * @param color color description object ( implementation of the IPalleteColor interface )
		 * @see com.yauthor.module.editor.services.colorpicker.IPalleteColor
		 **/
		function set selectedColor(color:IPalleteColor):void;
		

		/**
		 * Style property by which selected color was defined.
		 * @return color description object  
		 * @see com.yauthor.module.editor.services.colorpicker.IPalleteColor
		 **/
		
		function get selectedColorStyleProp():String;
		
		
		/**
		 * Method allows to set color selection by passing a style property.
		 * Setting this variable will automatically update 'selectedColor' property. 
		 * @param property a style property by which color was defined.
		 **/

		function set selectedColorStyleProp(property:String):void;
		
		
		/**
		 * Method returns reference to the 'ICssPalettePicker' visual component   
		 **/
		function get view():DisplayObject;
	}
}